SQUID - Sys 03 (copy 1)

A METHOD FOR SIMULATING A FLUX-LOCKED DC SQUID*

G.M. Gutt, N.J. Kasdin, M.R. Condron II, B. Muhlfelder, J.M. Lockhart†
W.W. Hansen Experimental Physics Laboratory, Stanford, CA 94305

M.W. Cromar
National Institute of Standards and Technology, Boulder, CO 80303

Abstract--Many high precision experiments place severe requirements on the noise, linearity and slew rate of flux-locked dc SQUID systems (linearity requirement approaches 1 in 10^6 for Gravity Probe-B). A computationally efficient and accurate method of simulating a dc SQUID's V-Φ and I-V characteristics has proven valuable in evaluating and improving various SQUID readout methods. The simulation of the SQUID is based on fitting of previously acquired data from either a real or a modeled device using the Fourier transform of the V-Φ curve. This method does not predict SQUID behavior, but rather is a way of replicating a known behavior efficiently with portability into various simulation programs such as SPICE. In this paper we discuss the methods used to simulate the SQUID and the flux-locking control electronics and present specific examples of this approach. Results include an estimate of the slew rate and linearity of a simple flux-locked loop using a characterized dc SQUID.

I. INTRODUCTION

In many dc SQUID (superconducting quantum interference device) applications it is useful to understand the interaction of the SQUID with external electronics such as transformers and flux-locking apparatus. Approximations based on a linear SQUID characteristic have been used in the past to predict many of these interactions but are often limited in their accuracy and scope due to the non-linear behavior of the SQUID [1]. Extensive modeling of intrinsic SQUID behavior, often employing the resistively shunted junction (RSJ) model predicts non-linear behavior [2,3]. Other simulation techniques attempt to quantify SQUID interactions with external electronics by using idealized SQUID behavior [4,5]. These simulation methods do not easily lend themselves to numerical circuit modeling and often fail to fully replicate real device performance. The technique we describe allows simulation of the response of actual SQUIDs based upon their measured non-ideal characteristics. Our simulator closely replicates the behavior of a characterized device over its full dc operating characteristic (V-Φ, I-V). These replications can then be easily imported into circuit simulation environments (including some implementations of SPICE) to accurately predict behavior such as slew rate, linearity and noise of flux-locked dc SQUID systems. An important assumption of this approach is that the measured SQUID characteristic remains unaffected by external circuitry.

Table 1. Frequently Used Symbols

Symbol / Variable	Meaning
Φ	Applied magnetic flux
Φ_0	Flux quantum (2.07×10^{-15} Wb)
Φ_{Lock}	Lock point in a flux-locked loop
Ib	dc bias current through SQUID
V	dc voltage across the SQUID
V'	Simulated dc SQUID voltage
V-Φ	Characteristic of V vs. Φ
I-V	Characteristic of V vs. Ib
Ic_{max}	Max. critical current of the SQUID
i	$(-1)^{1/2}$
Re{ }	Real part of { }
Im{ }	Imaginary part of { }

Table 1 lists the symbols and variables frequently used in this paper.

II. CHARACTERIZING THE DC SQUID

Before the SQUID simulation technique employed in this paper can be of use, one must first have the dc characterization for a device. In general this means that the V-Φ curve must be acquired at several different bias current settings.

Figure 1 shows data taken on a dc SQUID fabricated at the National Institute of Standards and Technology (NIST). The V-Φ curve was measured using a room temperature amplifier along with analog to digital (A/D) and digital to analog (D/A) converters. A computer controlled the data acquisition process and stored the results. Using an optically decoupled D/A and a large series resistor (10kΩ-100kΩ) a fixed bias current was set. A second D/A (with similar circuitry) then ramped current through the flux-modulation coil of the SQUID using a low frequency triangle wave. Enough current was used on the ramp to ensure several Φ_0 were observed at the SQUID output. The voltage across the SQUID was

*Research supported by NASA under contract NAS8-36125

†Also with Physics Dept., San Francisco State Univ.
Manuscript received August 24, 1992.



Fig. 1 Acquired V-Φ curve for a NIST dc SQUID. Ib=5μA to 30μA in 1.79μA steps. $1\Phi_0$=22.5μA through the flux-modulation coil.

amplified and then digitized by an A/D. The computer stored and averaged the information from each cycle of the triangle wave-form. By repeating the process for several bias currents a very clean and complete characterization was achieved.

III. CONDENSING THE CHARACTERISTIC

For a constant bias (Ib_1) the V-Φ curve can be conveniently modeled by using a truncated Fourier series; a finite series of sine and cosine terms with voltage as the dependent variable and flux as the independent variable, yielding:

$$V'(Ib_1,\Phi) = \sum_{K=0}^{M} [A_K \cos(2\pi K\Phi/\Phi_0)+B_K \sin(2\pi K\Phi/\Phi_0)] \quad (1)$$

where M is the number of harmonic terms and A_K, B_K are weighting coefficients for each harmonic term.

As M is increased, the fit of the model tends to increase in accuracy. Generally, the sharper a particular SQUID's transition from the zero voltage state to a positive (or negative) voltage, the larger M needs to be to accurately fit the characteristic. If the SQUID's characteristic contains severe discontinuities this method may not be appropriate due to the Gibbs phenomenon [6]. A benefit of this approach is that Eq. 1 produces a voltage that is periodic (with a $1\Phi_0$ period) and is defined for all Φ. A simple method of obtaining A_K and B_K is the discrete Fourier transform (DFT) or its functional equivalent (for powers of 2) the fast Fourier transform (FFT). If N equally spaced data points are taken over a period of $1\Phi_0$ at a fixed bias current, then the following formula can be used:

$$F(Ib_1,J)= N^{-1/2} \sum_{L=0}^{N-1} V(Ib_1,L\Delta)\, e^{-i2\pi JL/N} \quad (2)$$

where $\Delta=\Phi_0/N$ is the spacing in flux, $V(Ib_1,L\Delta)$ is the voltage at the flux state LΔ for a fixed bias Ib_1. $F(Ib_1,J)$ is the DFT of $V(Ib_1, L\Delta)$ with the frequency index J.

If Eq. 2 is used as the DFT algorithm then:

$$A_0=F(Ib_1,0)\, N^{-1/2} \qquad B_0=0$$
$$A_K=2\, Re\{F(Ib_1,K)\}\, N^{-1/2} \qquad B_K=-2\, Im\{F(Ib_1,K)\}\, N^{-1/2} \quad (3)$$

The variable N from Eq. 2 should be large enough to avoid aliasing [7]. Typically the harmonic content of a smooth dc SQUID can be represented with M=12 and often we choose N=128. Note that Eq. 3 holds only if the DFT formula of Eq. 2 is used. There are several definitions of the DFT (and FFT) which are similar except for a scaling factor.

Equation 3 represents the truncated Fourier series coefficients for a single current bias (Ib_1). When these coefficients are inserted into Eq. 1 a fit is constructed for all Φ for $Ib=Ib_1$. Assuming that V'-Φ data exists for several bias points it is advantageous to be able to approximate the curve between any two adjacent bias levels (Ib_1, Ib_2). This can be accomplished by linear, spline or polynomial interpolation between the matching A_K, B_K terms. In this paper we use the linear interpolation method. For example let $Ib_1< Ib_x<Ib_2$ then:

$$A_1(Ib_x)=[A_1(Ib_2)-A_1(Ib_1)]\, [Ib_x-Ib_1]/[Ib_2-Ib_1\,]+A_1(Ib_1) \quad (4)$$

where $A_1(Ib_x)$ is the coefficient A_1 for the bias current Ib_x.

In this manner a three dimensional mapping can be created for any dc SQUID with A's and B's on the Z axis, Ib on the Y axis, and the harmonic index (K) on the X axis. Thus, for any given Ib and Φ the SQUID voltage (V') can be approximated. For the bias current interpolation to work reasonably well we have found that approximately 15 bias currents should be used between Ib=0 and $Ib= 2Ic_{max}$ for a dc SQUID with a smooth V-Φ characteristic.

IV. CIRCUIT SIMULATIONS USING SPICE

PSpice is a commercial version of the standard circuit simulation program SPICE [8,9]. We used PSpice version 5.0 with the "Analog Behavioral Modeling" option. The strategy used to implement the SQUID model of section III in PSpice is now outlined.

1) Dummy nodes are set up using the "E" and "TABLE" statements. These nodes have large resistors to ground so that the circuit is closed and SPICE compiles the code. In this way nodes can be used as variables without affecting the non-SQUID circuit.

2) A dummy node is used to represent each A_x, B_x at all known bias values, for example:

E1 100 0 TABLE {V(1)}= $(Ib_1,A_1)(Ib_2,A_1)...(Ib_n,A_1)$

Based on the value of V(1), A_1 will be calculated using linear interpolation between the two closest Ib's. This value will be stored at node 100. Some scaling is done so that the voltage at node 1 represents Ib.

3) In the main section of the code where the SQUID voltage is to appear Eq. 1 is evaluated using the "E - VALUE" statement, for example:

ESQD 5 0 VALUE={V(100)*COS(2*PI*I(VFX))+...}

where PI (3.141...) is defined using the ".PARAM" statement, I(VFX) is scaled to be Φ/Φ_0 , and node 5 will exhibit the SQUID voltage.

A circuit external to the SQUID subcircuit is used to apply bias and flux to the SQUID. In practice the above code can be compressed using ".FUNC" statements. We use a computer program, written in Pascal, to take the SQUID characteristic (gathered in the manner discussed in section II) and automatically generate a SQUID subcircuit listing for SPICE.

V. RESULTS

We have used the techniques described to simulate SQUID behavior in complex (bias and flux modulation) flux-locking schemes. In general the agreement between our simulations and experimental results is better than 10% for slew rate and linearity. For brevity, we now describe simplified examples of our simulation technique.

Figure 2 is the simulated V'-Φ characteristic generated with the raw data shown in Fig. 1; the bias step from curve to curve (2μA) is different from that in Fig. 1 to show the results of interpolation. Fig. 3 is the simulated I-V' characteristic of the same data with Φ=0.

Linearity and slew rate of a simplified flux-locked system (no flux or bias modulation) is studied here with the use of a single-pole integrator. The SQUID voltage subtracted from a set voltage provides the integrator input. The output of the integrator causes negative feedback current to flow through the flux-modulation coil of the SQUID and attempts to maintain a constant flux state $\Phi=\Phi_{Lock}$.

To investigate linearity, a simulated pure $1\Phi_0$ (45μA p-p) 100Hz sinusoidal signal is injected into the flux-modulation coil of the flux-locked SQUID ($Ib=Ic_{max}$. Φ_{lock} is set so that V' is at a mid-point between the maximum and minimum of the V'-Φ curve). The closed loop bandwidth of the simulated system is about 1.6kHz. Ideally, the output of



Fig. 2 Simulated V'-Φ curve for NIST dc SQUID (M=11). Ib=8μA to 30μA in 2μA steps. Φ_0=22.5μA through the flux-modulation coil.



Fig. 3 Simulated I-V' curve for NIST dc SQUID (Φ=0) This curve is nearly indistinguishable from the I-V curve of the measured data.

the flux-locked system and the injected signal are identical (except for a scale factor and possible phase shift). In reality, non-linearities in the V-Φ curve produce non-linearities in the flux-locked output for finite open loop gain. Thus, the output of the flux-locked loop will contain harmonics of 100Hz. Fig 4. shows the FFT of the flux-locked output. The fundamental peak at 100Hz has been normalized so that the level on the graph represents the magnitude of the injected sinusoidal signal. In this example the first harmonic (200Hz) is a factor of 1000 smaller than the fundamental (linearity of 10^{-3}). By changing the SQUID bias point or the open loop gain, the non-linearities observed at the flux-locked output will vary.



Fig. 4 Normalized magnitude FFT of simulated flux-locked loop output. Input: $\Phi=1\Phi_0 \sin[2\pi(100t)]$.

Maximum slew-rate occurs at the value of $d\Phi/dt$ which first causes flux jumps in the output of the flux-locked loop, and is determined by the open loop gain of a flux-locked system. To quantify slew rate we inject a $1\Phi_0$ chirp signal into the flux-modulation coil of the simulated SQUID (see Fig. 5). At approximately time t=0.05 seconds the system loses lock and begins to jump (several flux jumps are observed). From the data in Fig. 5 along with the specifics of the injected chirp signal we find that the maximum slew rate is approximately $300\Phi_0$/sec (at 300Hz). A linear approximation for maximum slew rate is:

$$\text{Max}[d\Phi/dt]=[G(f)+1]\ f/4 \qquad (5)$$

where G(f) is the open loop gain at frequency f.



Fig. 5 Simulated flux-locked loop output. Input: $\Phi=1\Phi_0 \sin[2\pi(3000t^2)]$.

This assumes that $dV/d\Phi$ and G(f) are constant over $1/2\Phi_0$ for a fixed frequency. Due to the non-linearities of the V-Φ curve this is incorrect ($dV/d\Phi$ not constant) and often results in overestimates of the stability and slew rate limit; for the above example linear theory predicts a maximum slew rate of $440\Phi_0$/sec at 300Hz.

The previous simple examples differ from more typical (and complex) flux-locked systems in the following ways: they do not include bias or flux modulation, and they have relatively low open loop gain. The later generates results with lower maximum slew rate and greater non-linear behavior in the flux-locked output than is typically encountered in actual flux-locked systems.

CONCLUSIONS

We have demonstrated a computationally efficient and accurate method of simulating a dc SQUID in a flux locked loop. Tests of the NIST SQUID described here, operated with actual feedback electronics, gave distortion and slew rate limits in good agreement with this simulation method. Presently we are extending the SQUID simulation to include 1/f and white noise from various physical sources within the SQUID [10]. This will allow us to simulate complex flux and current modulation schemes (within a flux-locked loop) that are intended to reduce the effect of specific noise generators in the SQUID.

REFERENCES

[1] F. Wellstood, C. Heiden, and J. Clarke, "Integrated dc SQUID magnetometer with a high slew rate", Rev. Sci. Instrum., vol. 55 , pp. 952-957. June, 1984.

[2] C.D. Tesche, "A Thermal Activation Model for Noise in the DC SQUID", J. Low Temp. Phys.,vol. 44, pp. 119-147, 1981.

[3] V.J. deWall, P. Schrijner, R. Llurba, "Simulation and Optimization of a DC Squid with Finite Capacitance", J. Low Temp. Phys, vol. 54, pp. 215-232, 1984.

[4] A. Martinez, J. Flokstra, et al., "Low noise SQUID simulator with large dynamic range of up to eight flux quanta", Cryogenics, vol. 30, pp. 324-329, April 1990.

[5] R.W. Henry and D.E. Prober, "Electronic analogs of double-junction and single-junction SQUIDs", Rev. Sci. Instrum., vol. 52, pp. 902-914, June 1981.

[6] A. B. Carlson, *Communication Systems*, New York: McGraw-Hill, 1986, p. 29.

[7] *ibid.*, p. 353.

[8] "PSpice" and "Analog Behavioral Modeling" are registered trademark of MicroSim Corporation.

[9] We use trade names to specify the experimental procedure adequately and do not imply endorsement by NIST. Similar products by other manufacturers may work as well or better.

[10] N.J. Kasdin and T. Walter, "Discrete Simulation of Power Law Noise", *Proceedings of the 1992 Control Symposium*, to be published.

4

```
SQUID  An Illustration of a DC SQUID using PSpice

.OPT  ACCT LIST NODE NOPAGE VNTOL=1NV RELTOL=.0001 ITL1=500 ITL2=500 ITL4=500
.WIDTH OUT=80
.TRAN  100U .1 0 100U
.PROBE V(2) V(10) V(7)
.PARAM P2=6.2832

* FUNCTIONS CS(X) AND SN(X) ARE REQUIRED FOR SUBCKT SQUID1
.FUNC CS(X)  COS(P2*X/PHIO)
.FUNC SN(X)  SIN(P2*X/PHIO)

* EXTERNAL CIRCUIT ATTACHED TO SQUID
IBIAS  0  1  21U
VBIAS  1  2  0
IIN   0  6  0
EMOD  7  0  VALUE = {2*1.125*SIN(P2*100*TIME)}
RMOD  7  4  100K
XSQUID 6  4  2 SQUID1

* ESET1 SET THE VOLTAGE TO "LOCK IN" AND THE PREAMP GAIN
* VSET2 RAMPS VOLTAGE SO GAIN=0 AT T=0 (HELPS SPICE CONVERGE)
VSET2  16 0  PWL(0 0 1M 1 1 1)
RSET2  16 0  1MEG
ESET1  15 0  VALUE = {50K*V(16)*(V(2)-10U)}

RINOP  15 14  10K
EAMP1  10 0 0 14   100K
RLEAK  10 14  1MEG
CF1   10 14  0.01U
RFSQUD 10 4   100K

.SUBCKT SQUID1 1 10 20 PARAMS: PHIO=22.25U, INPHI=40.8, BSTP=1.786U, BSTT=5U, FLXOFF=8.63U

* NODE 1 INPUT COIL, NODE 10 MOD COIL, NODE 20 SQUID OUT, BIAS IN
* P2    = 2*PI
* PHIO  = MUTUAL INDUCTANCE THROUGH MOD COIL A/PHIO
* INPHI  = MULTIPLIER X TIMES MORE SENSITIVE TO CURRENT THAN MOD COIL
* BSTP   = BIAS STEP OF TABLE DATA
* BSTT   = STARTING BIAS OF TABLE DATA
* FLXOFF = MOD CURRENT AT WHICH FIRST PEAK OCCURS FOR SMALL BIAS

* INPUT COIL NODES 1 0
VIN   1  0  0

* MOD COIL INPUT NODES 10 0
VMOD  10 0  0

* EFLUX SUMS FLUX FROM MOD AND INPUT COILS, NORMALIZED TO MUTUAL INDUCTANCE
* THROUGH THE MOD COIL A/PHIO.

RFX   50 0 1MEG
GFX   50 51 VALUE = {V(41)*(I(VMOD)+40.8*I(VIN))+FLXOFF}
VFX   51 0  0

* BIAS INPUT/VOLTAGE OUTPUT NODES 20 0
VBSNS  20 21  0
EOUT   21 0   VALUE = {V(41)*V(30)}

* EOUT SOURCES COMPUTE THE ACTUAL OUTPUT VOLTAGE FOR THE SQUID USING SIN AND
* COS AND THE VALUES IN THE LOOKUP.
ROUT1  30 0   1MEG
EOUT1  30 31  VALUE = {V(101)+V(102)*CS(1*I(VFX))+V(103)*SN(1*I(VFX))}
EOUT2  31 32  VALUE = {     V(104)*CS(2*I(VFX))+V(105)*SN(2*I(VFX))}
EOUT3  32 33  VALUE = {     V(106)*CS(3*I(VFX))+V(107)*SN(3*I(VFX))}
EOUT4  33 34  VALUE = {     V(108)*CS(4*I(VFX))+V(109)*SN(4*I(VFX))}
EOUT5  34 35  VALUE = {     V(110)*CS(5*I(VFX))+V(111)*SN(5*I(VFX))}
EOUT6  35 36  VALUE = {     V(112)*CS(6*I(VFX))+V(113)*SN(6*I(VFX))}
EOUT7  36 37  VALUE = {     V(114)*CS(7*I(VFX))+V(115)*SN(7*I(VFX))}
EOUT8  37 38  VALUE = {     V(116)*CS(8*I(VFX))+V(117)*SN(8*I(VFX))}
EOUT9  38 39  VALUE = {     V(118)*CS(9*I(VFX))+V(119)*SN(9*I(VFX))}
EOUT10 39 0   VALUE = {     V(120)*CS(10*I(VFX))+V(121)*SN(10*I(VFX))}
```

```
* ECONV CONVERTS BIAS CURRENT FROM AMPS TO A NUMBER BETWEEN 1-15 FOR USE
* IN LOOK UP TABLES ECX.  5UA = 1, 6.786UA = 2 ECT. NUMBERS BELOW 1 SQUID OFF.
RCONV  40 0  1MEG
ECONV  40 0  VALUE =  {((ABS(I(VBSNS))-BSTT)/BSTP)+1}

* EMAG ALLOWS FOR A NEGATIVELY BIASED SQUID
RMAG   41 0  1MEG
EMAG   41 0  VALUE = {I(VBSNS)/ABS(I(VBSNS))}

* LOOK UP TABLE FOR SERIES COEFS, EC1=DC, EC2=COS FIRST HARM, EC3=SIN FIRST HARM
* EC4=COS SECOND HARMONIC ETC.. THIS MODEL USES DC PLUS 10 HARMONICS.

R101 101 0 1MEG
EC1  101 0 TABLE {V(40)}=
+ (-3,0)(1,1.021E-07)(2,6.766E-08)(3,6.863E-08)(4,2.100E-07)(5,4.714E-07)(6,1.126E-06)(7,2.310E-06)
+ (8,4.016E-06)(9,5.711E-06)(10,7.559E-06)(11,9.572E-06)(12,1.105E-05)(13,1.295E-05)(14,1.504E-05)(15,1.704E-05)

R102 102 0 1MEG
EC2  102 0 TABLE {V(40)}=
+ (-3,0)(1,3.630E-09)(2,-2.287E-09)(3,-2.671E-08)(4,-2.175E-07)(5,-7.055E-07)(6,-1.772E-06)(7,-3.397E-06)
+ (8,-5.261E-06)(9,-6.601E-06)(10,-7.061E-06)(11,-6.591E-06)(12,-5.360E-06)(13,-3.729E-06)(14,-2.161E-06)(15,-9.283E-07)

R103 103 0 1MEG
EC3  103 0 TABLE {V(40)}=
+ (-3,0)(1,-8.385E-10)(2,2.873E-09)(3,2.304E-08)(4,1.278E-07)(5,2.755E-07)(6,3.914E-07)(7,1.337E-07)
+ (8,-7.576E-07)(9,-2.157E-06)(10,-3.696E-06)(11,-5.099E-06)(12,-5.984E-06)(13,-6.119E-06)(14,-5.726E-06)(15,-5.247E-06)

R104 104 0 1MEG
EC4  104 0 TABLE {V(40)}=
+ (-3,0)(1,2.391E-09)(2,1.119E-09)(3,7.476E-09)(4,9.865E-08)(5,4.112E-07)(6,9.846E-07)(7,1.553E-06)
+ (8,1.750E-06)(9,1.510E-06)(10,1.183E-06)(11,7.186E-07)(12,4.211E-07)(13,3.532E-07)(14,2.418E-07)(15,8.129E-08)

R105 105 0 1MEG
EC5  105 0 TABLE {V(40)}=
+ (-3,0)(1,4.790E-10)(2,-2.641E-09)(3,-3.352E-08)(4,-1.821E-07)(5,-3.881E-07)(6,-5.087E-07)(7,-2.679E-07)
+ (8,2.394E-07)(9,6.016E-07)(10,8.017E-07)(11,7.076E-07)(12,2.899E-07)(13,6.378E-08)(14,1.242E-07)(15,1.948E-07)

R106 106 0 1MEG
EC6  106 0 TABLE {V(40)}=
+ (-3,0)(1,-4.438E-09)(2,1.398E-09)(3,1.281E-08)(4,9.657E-09)(5,-1.315E-07)(6,-3.291E-07)(7,-4.519E-07)
+ (8,-4.668E-07)(9,-4.929E-07)(10,-3.173E-07)(11,-8.886E-08)(12,1.754E-08)(13,1.045E-07)(14,1.013E-07)(15,4.428E-08)

R107 107 0 1MEG
EC7  107 0 TABLE {V(40)}=
+ (-3,0)(1,3.896E-09)(2,-5.626E-10)(3,2.425E-08)(4,1.526E-07)(5,3.311E-07)(6,3.815E-07)(7,2.591E-07)
+ (8,9.030E-08)(9,-1.175E-07)(10,-2.824E-07)(11,-1.195E-07)(12,3.632E-08)(13,9.829E-08)(14,1.255E-07)(15,1.239E-07)

R108 108 0 1MEG
EC8  108 0 TABLE {V(40)}=
+ (-3,0)(1,4.945E-10)(2,-2.699E-09)(3,-2.290E-08)(4,-6.189E-08)(5,-1.581E-08)(6,1.423E-08)(7,8.872E-08)
+ (8,2.033E-07)(9,2.456E-07)(10,9.884E-08)(11,2.806E-08)(12,1.778E-08)(13,-6.000E-09)(14,-1.980E-08)(15,-6.575E-09)

R109 109 0 1MEG
EC9  109 0 TABLE {V(40)}=
+ (-3,0)(1,-2.082E-09)(2,1.361E-09)(3,-6.613E-09)(4,-8.064E-08)(5,-1.851E-07)(6,-1.857E-07)(7,-1.576E-07)
+ (8,-8.178E-08)(9,1.068E-07)(10,1.263E-07)(11,2.423E-08)(12,-5.761E-09)(13,3.083E-08)(14,3.763E-08)(15,1.833E-08)

R110 110 0 1MEG
EC10 110 0 TABLE {V(40)}=
+ (-3,0)(1,1.323E-09)(2,-1.643E-10)(3,1.319E-08)(4,5.976E-08)(5,4.871E-08)(6,6.077E-08)(7,-3.343E-09)
+ (8,-1.116E-07)(9,-1.207E-07)(10,-3.851E-08)(11,-9.082E-09)(12,8.046E-10)(13,1.124E-08)(14,-5.394E-09)(15,-3.084E-09)

R111 111 0 1MEG
EC11 111 0 TABLE {V(40)}=
+ (-3,0)(1,1.598E-09)(2,-1.497E-09)(3,-7.099E-09)(4,2.272E-08)(5,6.239E-08)(6,5.406E-08)(7,8.038E-08)
+ (8,6.541E-08)(9,-3.910E-08)(10,-2.918E-08)(11,4.285E-08)(12,3.240E-08)(13,3.673E-08)(14,2.761E-08)(15,1.193E-08)

R112 112 0 1MEG
EC12 112 0 TABLE {V(40)}=
+ (-3,0)(1,5.671E-10)(2,-2.676E-09)(3,-4.195E-09)(4,-3.541E-08)(5,-2.292E-08)(6,-4.435E-08)(7,-1.933E-08)
+ (8,5.723E-08)(9,7.901E-08)(10,6.299E-09)(11,1.778E-08)(12,9.404E-10)(13,4.105E-09)(14,1.680E-09)(15,3.283E-09)
```

```
R113 113 0 1MEG
EC13 113 0 TABLE {V(40)}=
+ (-3,0)(1,7.250E-10)(2,2.750E-09)(3,1.172E-08)(4,6.615E-09)(5,-6.337E-09)(6,3.982E-09)(7,-4.282E-08)
+ (8,-6.028E-08)(9,3.657E-08)(10,4.107E-08)(11,7.449E-09)(12,1.368E-08)(13,2.364E-08)(14,1.732E-08)(15,1.849E-08)

R114 114 0 1MEG
EC14 114 0 TABLE {V(40)}=
+ (-3,0)(1,2.245E-09)(2,2.629E-09)(3,9.135E-10)(4,1.160E-08)(5,-4.202E-09)(6,2.184E-08)(7,2.629E-08)
+ (8,-2.665E-08)(9,-4.638E-08)(10,4.940E-09)(11,-2.716E-09)(12,8.558E-09)(13,8.917E-09)(14,2.895E-10)(15,1.574E-09)

R115 115 0 1MEG
EC15 115 0 TABLE {V(40)}=
+ (-3,0)(1,1.215E-10)(2,1.795E-09)(3,-6.936E-09)(4,-1.509E-08)(5,-3.418E-09)(6,-2.069E-08)(7,2.595E-08)
+ (8,5.170E-08)(9,-3.255E-08)(10,-4.591E-09)(11,8.217E-09)(12,1.990E-08)(13,1.946E-08)(14,1.955E-08)(15,1.333E-08)

R116 116 0 1MEG
EC16 116 0 TABLE {V(40)}=
+ (-3,0)(1,2.433E-09)(2,1.570E-09)(3,4.007E-09)(4,-2.043E-09)(5,1.162E-08)(6,-5.830E-09)(7,-2.924E-08)
+ (8,1.385E-08)(9,2.746E-08)(10,1.352E-08)(11,-5.683E-09)(12,4.159E-09)(13,8.678E-10)(14,1.994E-09)(15,2.330E-09)

R117 117 0 1MEG
EC17 117 0 TABLE {V(40)}=
+ (-3,0)(1,2.623E-09)(2,7.249E-10)(3,7.134E-10)(4,6.850E-09)(5,-6.703E-09)(6,1.937E-08)(7,-1.416E-08)
+ (8,-4.225E-08)(9,3.333E-08)(10,2.030E-08)(11,1.402E-08)(12,1.641E-08)(13,2.057E-08)(14,1.624E-08)(15,1.047E-08)

R118 118 0 1MEG
EC18 118 0 TABLE {V(40)}=
+ (-3,0)(1,-2.304E-09)(2,-1.370E-09)(3,-3.247E-09)(4,-1.307E-09)(5,-5.208E-09)(6,-3.130E-09)(7,3.037E-08)
+ (8,-1.332E-08)(9,-1.414E-08)(10,-1.106E-09)(11,3.425E-09)(12,2.770E-09)(13,3.970E-09)(14,1.644E-09)(15,1.009E-09)

R119 119 0 1MEG
EC19 119 0 TABLE {V(40)}=
+ (-3,0)(1,-2.704E-09)(2,-1.740E-09)(3,1.279E-09)(4,-4.905E-09)(5,1.100E-08)(6,-1.442E-08)(7,5.298E-09)
+ (8,3.061E-08)(9,-2.254E-08)(10,-5.584E-09)(11,1.844E-08)(12,1.099E-08)(13,1.455E-08)(14,1.312E-08)(15,1.316E-08)

R120 120 0 1MEG
EC20 120 0 TABLE {V(40)}=
+ (-3,0)(1,-1.213E-11)(2,-3.136E-09)(3,-2.699E-10)(4,-1.017E-09)(5,-2.250E-09)(6,4.751E-09)(7,-1.827E-08)
+ (8,7.246E-09)(9,1.218E-08)(10,-3.066E-10)(11,5.242E-09)(12,2.822E-09)(13,3.317E-09)(14,1.514E-09)(15,-1.482E-09)

R121 121 0 1MEG
EC21 121 0 TABLE {V(40)}=
+ (-3,0)(1,2.038E-09)(2,-1.736E-09)(3,-4.591E-09)(4,1.931E-09)(5,-1.338E-08)(6,8.826E-09)(7,-2.540E-09)
+ (8,-1.304E-08)(9,2.259E-08)(10,7.815E-09)(11,1.123E-08)(12,1.177E-08)(13,1.228E-08)(14,1.292E-08)(15,8.057E-09)

.ENDS
.END
```